chapter

Op-Amp Circuits 10

10-1 INTRODUCTION

An **op-amp** may be modeled as a linear amplifier to simplify the design and analysis of op-amp circuits. The linear models give reasonable results, especially for determining the approximate design values of op-amp circuits. However, the simulation of the actual behavior of op-amps is required in many applications to obtain accurate responses for the circuits. PSpice does not have any model for op-amps. However, an op-amp can be simulated from the circuit arrangement of the particular type of op-amp. The μA741 type of op-amp consists of 24 transistors, and it is beyond the capability of the student (or demo) version of PSpice. However, a macromodel, which is a simplified version of the op-amp and requires only two transistors, is quite accurate for many applications and can be simulated as a subcircuit or library file. Some manufacturers often supply macromodels of their op-amps [1]. In the absence of a complex op-amp model, the characteristics of op-amp circuits may be determined approximately by one of the following models:

Dc linear model
Ac linear model
Nonlinear macromodel

10-2 DC LINEAR MODELS

An op-amp may be modeled as a voltage-controlled voltage source, as shown in Fig. 10-1(a). The input resistance is high, typically 2 MΩ, and the output resistance is very low, typically 75 Ω. For an ideal op-amp, the model in Fig. 10-1(a)



(a) Dc model

(b) Simple Dc model

Figure 10-1 Dc linear models.

can be reduced to that of Fig. 10-1(b). These models do not take into account the saturation effect and slew rate, which do exist in actual op-amps. The gain is also assumed to be independent of the frequency, but the gain of actual practical op-amps falls with the frequency. These simple models are normally suitable for dc or low-frequency applications.

10-3 AC LINEAR MODEL

The frequency response of an op-amp can be approximated by a single break frequency, as shown in Fig. 10-2(a). This characteristic can be modeled by the circuit of Fig. 10-2(b). This is a high-frequency model of op-amps. If an op-amp has more than one break frequency, it can be represented by using as many capacitors as the number of breaks. R_i is the input resistance and R_o is the output resistance.



(a) Frequency response



(b) Circuit model

Figure 10-2 Ac linear model with a single break frequency.

The dependent sources of the op-amp model in Fig. 10-2(b) have a common node. Without this, PSpice will give an error message because there is no dc path from the nodes of the dependent current source. The common node could be either with the input stage or with the output stage. This model does not take into account the saturation effect and is suitable only if the op-amp operates within the linear region.

The output voltage can be expressed as

$$V_{out} = -A_0 V_2 = \frac{-A_0 V_{in}}{1 + R_1 C_1 s}$$

Substituting $s = j2\pi f$ yields

$$V_{out} = \frac{-A_0 V_{in}}{1 + j2\pi f R_1 C_1} = \frac{-A_0 V_{in}}{1 + jf/f_b}$$

where

$f_b = 1/(2\pi R_1 C_1)$ is called the *break frequency,* in hertz.
A_0 = the *large-signal* (or *dc*) *gain* of the op-amp.

Thus, the open-loop voltage gain is

$$A(f) = \frac{V_{out}}{V_{in}} = -\frac{A_0}{1 + jf/f_b}$$

For μA741 op-amps, $f_b = 10$ Hz, $A_0 = 2 \times 10^5$, $R_i = 2\ \text{M}\Omega$, and $R_o = 75\ \Omega$. Letting $R_1 = 10\ \text{k}\Omega$, $C_1 = 1/(2\pi \times 10 \times 10 \times 10^3) = 1.15619\ \mu\text{F}$.

10-4 NONLINEAR MACROMODEL

The circuit arrangement of the **op-amp macromodel** is shown in Fig. 10-3 [1, 2, 3]. The macromodel can be used as a subcircuit with the .SUBCKT command. However, if an op-amp is used in various circuits, it is convenient to have the macromodel as a library file, namely, EVAL.LIB, and it is not required to type the statements of the macromodel in every circuit where the macromodel is employed. The library file EVAL.LIB that comes with the student version of PSpice has macromodels for op-amps, comparators, diodes, MOSFETs, BJTs, and SCRs. The macromodels for the linear operational amplifier of type LM324, the linear operational amplifier of type μA741, and the voltage comparator of type LM111 are included in the EVAL.LIB file. The professional version of PSpice supports library files for many devices.

The macromodel of the μA741 op-amp is simulated at room temperature. The library file EVAL.LIB contains the op-amp macromodel model as a subcircuit definition μA741 with a set of .MODEL statements. This op-amp model contains nominal, not worst-case, devices, and does not consider the effects of temperature.



Figure 10-3 Circuit diagram of op-amp macromodel.

The listing of the library file, EVAL.LIB, follows.

```
* Library file "EVAL.LIB" for UA741 op-amp
* connections:    noninverting input
*                 :   inverting input
*                 :   :
*                 :   :        positive power supply
*                 :   :        :   negative power supply
*                 :   :        :   :   output
*                 :   :        :   :   :
.SUBCKT UA741     1   2        4   5   6
*                 Vi+ Vi- Vp+ Vp- Vout
Q1   7   1   10 UA741QA
Q2   8   2   9  UA741QB
RC1  4   7   5.305165D+03
RC2  4   8   5.305165D+03
C1   7   8   5.459553D-12
RE1  10  11  2.151297D+03
RE2  9   11  2.151297D+03
IEE  11  5   1.666000D-05
```

```
CE    11   0    3.000000D-12
RE    11   0    1.200480D+07
GCM   0    12   11   3    5.960753D-09
GA    12   0    8    7    1.884955D-04
R2    12   0    1.000000D+05
C2    12   13   3.000000D-11
GB    13   0    12   0    2.357851D+02
RO2   13   0    4.500000D+01
D1    13   14   UA741DA
D2    14   13   UA741DA
EC    14   0    6    3    1.0
RO1   13   6    3.000000D+01
D3    6    15   UA741DB
VC    4    15   2.803238D+00
D4    16   6    UA741DB
VE    16   5    2.803238D+00
RP    4    5    18.16D+03
*  Models for diodes and transistors
.MODEL UA741DA D (IS=9.762287D-11)
.MODEL UA741DB D (IS=8.000000D-16)
.MODEL UA741QA NPN (IS=8.000000D-16 BF=9.166667D+01)
.MODEL UA741QB NPN (IS=8.309478D-16 BF=1.178571D+02)
* End of library file
*  End of subcircuit definition
.ENDS
```

Example 10-1

An inverting amplifier is shown in Fig. 10-4. The output is taken from node 5. Calculate and print the voltage gain, the input resistance, and the output resistance. The op-amp, which is modeled by the circuit in Fig. 10-1(a), has $A_0 = 2 \times 10^5$, $R_i = 2\ \text{M}\Omega$, and $R_o = 75\ \Omega$.



Figure 10-4 Inverting amplifier.

Solution The listing of the subcircuit file follows.

Example 10-1 Inverting amplifier

```
▲ *   Input voltage is 1.5 V dc.
  VIN  1 0  DC  1.5V
```

```
▲▲ R1    2  3  10K
   R2    4  0  6.67K
   RF    3  5  20K
   *  Calling subcircuit OPAMP
   XA1  2  1  2  0  OPAMP
   XA2  3  4  5  0  OPAMP
   *  Subcircuit definition for OPAMP
   .SUBCKT OPAMP 1  2  5  4
   RI  1  2  2MEG
   RO  3  5  75
   *  Voltage-controlled voltage source with a gain of 2E+5. The polarity of
   *  the output voltage is taken into account by changing the location of
   *  the controlling nodes.
   EA  3  4  2  1  2E+5
   *  End of subcircuit definition
   .ENDS    OPAMP
▲▲▲ *   Transfer-function analysis calculates and prints the dc gain,
    *   the input resistance, and the output resistance.
    .TF  V(5)  VIN
.END
```

The results of the transfer function analysis by the .TF command are given below:

```
 ****      SMALL-SIGNAL BIAS SOLUTION                 TEMPERATURE =  27.000 DEG C
 NODE     VOLTAGE     NODE     VOLTAGE     NODE     VOLTAGE     NODE      VOLTAGE
(    1)    1.5000    (    2)   1.5000    (    3)   15.11E-06   (    4)   50.21E-09
(    5)   -2.9999    (XA1.3)   1.5112    (XA2.3)    -3.0112
    VOLTAGE SOURCE CURRENTS
    NAME          CURRENT
    VIN          -3.778E-12
    TOTAL POWER DISSIPATION  5.67E-12  WATTS

 ****      SMALL-SIGNAL CHARACTERISTICS
      V(5)/VIN = -2.000E+00
      INPUT RESISTANCE AT VIN = 3.970E+11
      OUTPUT RESISTANCE AT V(5) = 1.132E+03
          JOB CONCLUDED
          TOTAL JOB TIME           2.42
```

Example 10-2

An integrator circuit is shown in Fig. 10-5(a). For the input voltage as shown in Fig. 10-5(b), plot the transient response of the output voltage for a duration of 0 to 4 ms in steps of 50 μs. The op-amp that is modeled by the circuit in Fig. 10-2(b) has R_i = 2 MΩ, R_o = 75 Ω, C_1 = 1.5619 μF, R_1 = 10 kΩ, and $A_0 = 2 \times 10^5$.



Figure 10-5 Integrator circuit.

Solution The listing of the circuit file follows:

Example 10-2 Integrator circuit

```
▲ *  The input voltage is represented by a piecewise linear waveform.
  *  To avoid convergence problems due to a rapid change, the input
  *  voltage is assumed to have a finite slope.
  VIN  1  0  PWL (0  0 1NS -1V 1MS -1V  1.0001MS 1V 2MS 1V
  +    2.0001MS -1V 3MS -1V  3.0001MS 1V  4MS 1V)
▲▲ R1  1  2  2.5K
   RF  2  4  1MEG
   RX  3  0  2.5K
   RL  4  0  100K
   C1  2  4  0.1UF
   *  Calling subcircuit OPAMP
   XA1  2  3  4  0  OPAMP
   *  Subcircuit definition for OPAMP
   .SUBCKT OPAMP  1  2  7  4
   RI  1  2  2.0E6
   *  Voltage-controlled current source with a gain of 1
   GB  4  3  1  2  0.1M
   R1  3  4  10K
   C1  3  4  1.5619UF
   *  Voltage-controlled voltage source with a gain of 2E+5
   EA  4  5  3  4  2E+5
   RO  5  7  75
   *  End of subcircuit OPAMP
   . ENDS
▲▲▲ *  Transient analysis for 0 to 4 ms with 50-μs increment
    .TRAN  50US  4MS
    *  Plot the results of transient analysis
    .PLOT  TRAN  V(4)  V(1)
    .PLOT  AC  VM(4)  VP(4)
    .PROBE
.END
```

The transient response for Example 10-2 is shown in Fig. 10-6. The .PLOT statements generate graphical plots in the output file. If the .PROBE command is included, there is no need for the .PLOT commands.



Figure 10-6 Transient response for Example 10-2.

Example 10-3

A practical differentiator circuit is shown in Fig. 10-7(a). For the input voltage as shown in Fig. 10-7(b), plot the transient response of the output voltage for a duration of 0 to 4 ms in steps of 50 μs. The op-amp, which is modeled by the circuit in Fig. 10-2(b), has $R_i = 2\ \text{M}\Omega$, $R_o = 75\ \Omega$, $C_1 = 1.5619\ \mu\text{F}$, $R_1 = 10\ \text{k}\Omega$, and $A_0 = 2 \times 10^5$.



Figure 10-7 Differentiator circuit.

Solution The listing of the circuit file follows.

Example 10-3 Differentiator circuit

```
▲ *   The maximum number of points is changed to 410. The default
  *   value is only 201.
  .OPTIONS  NOPAGE  NOECHO LIMPTS=410
  *  Input voltage is a piecewise linear waveform for transient analysis.
  VIN 1  0  PWL (0  0  1MS  1  2MS  0  3MS  1  4MS  0)
▲▲ R1  1  2  100
   RF  3  4  10K
   RX  5  0  10K
   RL  4  0  100K
   C1  2  3  0.4UF
   *  Calling op-amp OPAMP
   XA1  3  5  4  0  OPAMP
   *  Op-amp subcircuit definition
   .SUBCKT  OPAMP  1  2  7  4
   RI  1  2  2.0E6
   *  Voltage-controlled current source with a gain of 0.1M
   GB  4  3  1  2  0.1M
   R1  3  4  10K
   C1  3  4  1.5619UF
   *  Voltage-controlled voltage source with a gain of 2E+5
   EA  4  5  3  4  2E+5
   RO  5  7  75
   *  End of subcircuit OPAMP
   .ENDS  OPAMP
▲▲▲ *  Transient analysis for 0 to 4 ms with 50 μs increment
    .TRAN  10US  4MS
    *  Plot the results of transient analysis 4
    .PLOT  TRAN  V(4)  V(1)
    .PROBE
.END
```

The transient response for Example 10-3 is shown in Fig. 10-8. If the .PROBE command is included, there is no need for the .PLOT command.

Example 10-4

A filter circuit is shown in Fig. 10-9. Plot the frequency response of the output voltage. The frequency is varied from 10 Hz to 100 MHz with an increment of 1 decade and 10 points per decade. For the op-amp modeled by the circuit in Fig. 10-2(b), $R_i = 2\ \text{M}\Omega$, $R_o = 75\ \Omega$, $C_1 = 1.5619\ \mu\text{F}$, $R_1 = 10\ \text{k}\Omega$, and $A_0 = 2 \times 10^5$.

Solution The listing of the circuit file follows.

Example 10-4 A filter circuit

```
▲ *   Input voltage is 1 V peak for ac analysis or frequency response.
  VIN  1  0  AC  1
  R1  1  2  20K
  R2  2  4  20K
  R3  3  0  10K
  R4  1  5  10K
```



Figure 10-8 Transient response for Example 10-3.



Figure 10-9 A filter circuit.

```
    R5  4  5  10K
    R6  6  7  100K
    RL  7  0  100K
    C1  2  4  0.01UF
▲▲ *  Subcircuit call for OPAMP
     XA1  2  3  4  0  OPAMP
     XA2  5  6  7  0  OPAMP
     *  Subcircuit definition for OPAMP
     .SUBCKT  OPAMP  1  2  7  4
     RI  1  2  2.0E6
     *  Voltage-controlled current source with a gain of 0.1M
     GB  4  3  1  2  0.1M
     R2  3  4  10K
     C2  3  4  1.5619UF
     *  Voltage-controlled voltage source of gain 2E+5
     EA  4  5  3  4  2E+5
     RO  5  7  75
     *  End of subcircuit definition
     .ENDS  OPAMP
▲▲▲ *  AC analysis for 10 Hz to 100 MHz with a decade increment and
       *  10 points per decade
       .AC  DEC  10  10HZ  100MEGHZ
       *  Plot the results of ac analysis
       .PLOT  AC  VM(7)  VP(7)
       .PROBE
.END
```

The frequency response for Example 10-4 is shown in Fig. 10-10. If the .PROBE command is included, there is no need for the .PLOT command.

Example 10-5

A band-pass active filter is shown in Fig. 10-11. The op-amp can be modeled as a macromodel, as shown in Fig. 10-3. The description of the UA741 macromodel is listed in the library file EVAL.LIB. Plot the frequency response if the frequency is varied from 100 Hz to 1 MHz with an increment of 1 decade and 10 points per decade. The peak input voltage is 1 V.

Solution The listing of the circuit file follows.

Example 10-5 Band-pass active filter

```
▲ *  Input voltage of 1 V peak for frequency response
   VIN  1  0  AC  1
▲▲ R1  1  2  5K
    R2  3  4  1.5K
    R3  2  0  265K
    C1  2  4  0.01UF
    C2  2  3  0.01UF
    RL  4  0  15K
    VCC  6  0  DC  12V
    VEE  0  7  DC  12V
    *  Subcircuit call for UA741
```



Figure 10-10 Frequency response for Example 10-4.



Figure 10-11 Band-pass active filter.

```
    X1  0    3    6    7  4    UA741
    *  Vi+  Vi-  Vp+  Vp-  Vout
    *  Call library file EVAL.LIB
    .LIB  EVAL.LIB
▲▲▲ *   AC analysis for 100 Hz to 1 MHz with a decade increment and 10
      *   points per decade
      .AC  DEC  10  100HZ  1MEGHZ
      *  Plot the results of the ac analysis: magnitude of voltage at node 4
      .PLOT  AC  VM(4)
      .PROBE
.END
```

The frequency response for Example 10-5 is shown in Fig. 10-12. If the .PROBE command is included, there is no need for the .PLOT command.



Figure 10-12 Frequency response for Example 10-5.

Example 10-6

A free-running multivibrator circuit is shown in Fig. 10-13. Plot the transient response of the output voltage for a duration of 0 to 4 ms in steps of 20 μs. The op-amp can be modeled as a macromodel as shown in Fig. 10-3. The description of the UA741 macromodel is listed in library file EVAL.LIB. Assume the initial voltage of the capacitor $C_1 = -5$ V.



Figure 10-13 Free-running multivibrator.

Solution The listing of the circuit file follows.

Example 10-6 Free-running multivibrator

```
▲ VCC  6  0  DC  12V
  VEE  0  7  DC  12V
▲▲ R1  1  0  100K
   R2  1  2  100K
   R3  2  3  10K
   C1  3  0  0.1UF  IC=-5V
   *  Subcircuit call for UA741
   XA1  1    3    6   7    2    UA741
   *    Vi+  Vi-  Vp+  Vp-  Vout
   *  Call library file EVAL.LIB
   .LIB  EVAL.LIB
▲▲▲ *  Transient analysis from 0 to 4 ms in steps of 20 μs
    .TRAN  10US  4MS  UIC
    .PROBE
.END
```

The transient response for Example 10-6 is shown in Fig. 10-14.

Example 10-7

The circuit diagram of a differential amplifier with a transistor current source is shown in Fig. 10-15. Calculate the dc voltage gain, the input resistance, and the output resistance. The input voltage is 0.1 V. The model parameters of the bipolar transistors are BF=50, RB=70, and RC=40.

Solution The listing of the circuit file follows.

Example 10-7 Differential amplifier

```
▲ VCC  11  0   12V
  VEE  0   10  12V
```



Figure 10-14 Transient response for Example 10-6.



Figure 10-15 Differential amplifier.

```
   VIN  1   0  DC   0.25V
▲▲ RC1  11   3   10K
   RC2  11   5   10K
   RE1  4   12   150
   RE2  7   12   150
   RS1  1   2    1.5K
   RS2  6   0    1.5K
   RX   11   8   20K
   *   Model for NPN BJTs with model name QN
   .MODEL QN NPN (BF=50 RB=70 RC=40)
   Q1   3   2   4  QN
   Q2   5   6   7  QN
   Q3   12   8   9 QN
   Q4   9   9   10 QN
   Q5   8   9   10 QN
▲▲▲ *  DC transfer function analysis
    .TF  V(3,5)  VIN
.END
```

The results of the transfer-function analysis by the .TF commands are given below:

```
****     SMALL-SIGNAL BIAS SOLUTION                    TEMPERATURE =  27.000 DEG C
 NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE
(    1)       .2500    (    2)       .2190    (    3)     1.6609    (    4)      -.5575
(    5)     11.3460    (    6)      -.0020    (    7)     -.7057    (    8)    -10.4430
(    9)    -11.2220    (   10)    -12.0000    (   11)    12.0000    (   12)      -.7157
   VOLTAGE SOURCE CURRENTS
   NAME          CURRENT
   VCC          -2.221E-03
   VEE          -2.243E-03
   VIN          -2.068E-05
   TOTAL POWER DISSIPATION   5.36E-02   WATTS

****     SMALL-SIGNAL CHARACTERISTICS
     V(3,5)/VIN = -2.534E+01
     INPUT RESISTANCE AT VIN = 3.947E+04
     OUTPUT RESISTANCE AT V(3,5) = 2.000E+04
          JOB CONCLUDED
          TOTAL JOB TIME            4.01
```

REFERENCES

1. *Linear Circuits—Operational Amplifier Macromodels.* Dallas, Texas: Texas Instruments, 1990.
2. G. Boyle, B. Cohn, D. Pederson, and J. Solomon, "Macromodeling of integrated circuit operational amplifiers," *IEEE Journal of Solid-State Circuits,* Vol. SC-9, No. 6, December 1974, pp. 353–364.

3. I. Getreu, A. Hadiwidjaja, and J. Brinch, "An integrated-circuit comparator macromodel," *IEEE Journal of Solid-State Circuits,* Vol. SC-11, No. 6, December 1976, pp. 826–833.

4. S. Progozy, "Novel applications of SPICE in engineering education," *IEEE Transactions on Education,* Vol. 32, No. 1, February 1990, pp. 35–38.

PROBLEMS

10-1. Plot the frequency response of the integrator in Fig. 10-5 if the frequency is varied from 10 Hz to 100 kHz with a decade increment and 10 points per decade. The peak input voltage is 1 V.

10-2. Plot the frequency response of the differentiator in Fig. 10-7 if the frequency is varied from 10 Hz to 100 kHz with a decade increment and 10 points per decade. The peak input voltage is 1 V.

10-3. Repeat Example 10-2 if the macromodel of the op-amp in Fig. 10-3 is used. The supply voltages are V_{CC} = 15 V and V_{EE} = −15 V.

10-4. Repeat Example 10-3 if the macromodel of the op-amp in Fig. 10-3 is used. The supply voltages are V_{CC} = 15 V and V_{EE} = −15 V.

10-5. A full-wave precision rectifier is shown in Fig. P10-5. If the input voltage is v_{in} = 0.1 sin(2000πt), plot the transient response of the output voltage for a duration of 0 to 1 ms in steps of 10 μs. The op-amp can be modeled by the circuit of Fig. 10-2(b), and has R_i = 2 MΩ, R_o = 75 Ω, C_1 = 1.5619 μF, R_1 = 10 kΩ, and A_o = 2 × 10^5. Use the default values for the diode model. The supply voltages are V_{CC} = 12 V and V_{EE} = 12 V.



Figure P10-5

10-6. For Fig. P10-5, plot the dc transfer characteristics. The input voltage is varied from −1 V to 1 V in steps of 0.01 V.

10-7. For Fig. P10-7, plot the dc transfer characteristics. The input voltage is varied from −10 V to 10 V in steps of 0.1 V. The op-amp can be modeled as a macromodel, as shown in Fig. 10-3. The description of the macromodel is listed in library file EVAL.LIB. Use the default values for the diode model.



Figure P10-7

10-8. For Fig. P10-8, plot the dc transfer function. The input voltage is varied from −10 V to 10 V in steps of 0.1 V. The Zener voltages are $V_{Z1} = V_{Z2} = 6.3$ V. The op-amp can be modeled as a macromodel, as shown in Fig. 10-3. The description of the macromodel is listed in library file EVAL.LIB. The dc supply voltages of the op-amp are $V_{CC} = |V_{EE}| = 12$ V.



Figure P10-8

10-9. An integrator circuit is shown in Fig. P10-9(a). For the input voltage as shown in Fig. P10-9(b), calculate the slew rate of the amplifier by plotting the transient response of the output voltage for a duration of 0 to 200 μs in steps of 2 μs. For the op-amp modeled by the circuit in Fig. 10-2(b), $R_i = 2\ M\Omega$, $R_0 = 75\ \Omega$, $C_1 = 1.5619$ μF, $R_1 = 10\ k\Omega$, and $A_0 = 2 \times 10^5$.



(a) Circuit (b) Input voltage

Figure P10-9

10-10. Repeat Problem 10-9 if the macromodel of the op-amp in Fig. 10-3 is used. The supply voltages are $V_{CC} = 12$ V and $V_{EE} = -12$ V.

10-11. A sine-wave oscillator is shown in Fig. P10-11. Plot the transient response of the output voltage for a duration of 0 to 2 ms in steps of 0.1 ms. The op-amp can be modeled by the circuit of Fig. 10-2(b), and it has $R_i = 2\ M\Omega$, $R_o = 75\ \Omega$, $C_1 = 1.5619$ μF, $R_1 = 10\ k\Omega$, and $A_0 = 2 \times 10^5$.



Figure P10-11

10-12. For the gyrator in Fig. P10-12, plot the frequency response of the input impedance. The frequency is varied from 10 Hz to 10 MHz with a decade increment and 10 points per decade. For the op-amp modeled by the circuit in Fig. 10-2(b), $R_i = 2$ $M\Omega$, $R_o = 75\ \Omega$, $C_1 = 1.5619\ \mu F$, $R_1 = 10\ k\Omega$, and $A_0 = 2 \times 10^5$.

10-13. Use PSpice to perform a Monte Carlo analysis for five runs and for the dc analysis of Example 10-7. The output voltage is taken between nodes 3 and 5. The model parameter is R=1 for resistors. The lot deviation for all resistances is ±15%. The transistor parameter having uniform deviations is

$$BF = 50 \pm 20$$

(a) The greatest difference of the output voltage from the nominal run is to be printed.
(b) The maximum value of the output voltage is to be printed.
(c) The minimum value of the output voltage is to be printed.



Figure P10-12

10-14. Use PSpice to perform the worst-case analysis for Problem 10-13.

10-15. Use PSpice to perform a Monte Carlo analysis for six runs and for the transient response of Problem 10-5. The model parameter is $R=1$ for resistors. The lot deviation for all resistances is $\pm 20\%$.
(a) The greatest difference of the output from the nominal run is to be printed.
(b) The maximum value of the output voltage is to be printed.
(c) The minimum value of the output voltage is to be printed.

10-16. Use PSpice to perform the worst-case analysis for Problem 10-15.

10-17. Use PSpice to perform a Monte Carlo analysis for five runs and for the dc response of Problem 10-7. The model parameter is $R=1$ for resistors. The lot deviation for all resistances is $\pm 15\%$. The diode parameters having uniform deviations are

$$V_{Z1} = V_{Z2} = 6.3\text{V} \pm 1.3\text{V}$$

(a) The greatest difference of the output voltage from the nominal run is to be printed.
(b) The maximum value of the output voltage is to be printed.
(c) The minimum value of the output voltage is to be printed.

10-18. Use PSpice to perform the worst-case analysis for Problem 10-17.

10-19. Use PSpice to perform a Monte Carlo analysis for five runs and for the dc response of Problem 10-8. The model parameter is $R=1$ for resistors. The lot deviation for all resistances is $\pm 15\%$. The diode parameters having uniform deviations are

$$V_{Z1} = V_{Z2} = 6.3\text{V} \pm 1.3\text{V}$$

(a) The greatest difference of the output voltage from the nominal run is to be printed.
(b) The maximum value of the output voltage is to be printed.
(c) The minimum value of the output voltage is to be printed.

10-20. Use PSpice to perform the worst-case analysis for Problem 10-19.

10-21. Use PSpice to perform a Monte Carlo analysis for five runs and for the transient response of Problem 10-11. The model parameter is $R=1$ for resistors, and $C = 1$ for capacitors. The lot deviations for all resistances and capacitances are $\pm 15\%$.

(a) The greatest difference of the output voltage from the nominal run is to be printed.
(b) The maximum value of the output voltage is to be printed.
(c) The minimum value of the output voltage is to be printed.

10-22. Use PSpice to perform the worst-case analysis for Problem 10-21.